

SOLUTIONS OF FIRST ORDER DIFFERENTIAL EQUATIONS

```
1 '=====  
2 ' SOLUTIONS OF FIRST ORDER DIFFERENTIAL EQUATIONS  
3 '=====  
10 DEF FN F(X,Y)=Y-X:A$="y-x"  
20 MODE 2:LOCATE 13,4  
30 PRINT "APROXIMATE SOLOUTIONS FOR DIFERRENTIAL EQUATIONS"  
40 PLOT 10,10:DRAW 10,390:DRAW 630,390:DRAW 630,10:DRAW 10,10  
50 PLOT 5,5:DRAW 5,395:DRAW 635,395:DRAW 635,5:DRAW 5,5  
60 PLOT 10,300:DRAW 630,300:PLOT 10,80:DRAW 630,80  
70 WINDOW #1,23,60,9,20  
80 PRINT #1,"1. EULER APROXIMATE METHOD"  
90 PRINT #1,"2. RUNGE-KUTTA METHOD"  
100 PRINT #1,"3. HEUN METHOD"  
110 PRINT #1,"4. NYSTROM METHOD"  
120 PRINT #1,"5. RUNGE-KUTTA-GILL METHOD"  
130 PRINT #1,"6. MILNE METHOD"  
140 PRINT #1,"7. HAMMING METHOD"  
150 PRINT #1,"8. ADAMS-MULTON METHOD"  
160 PRINT #1,"9. MODIFIED MILNE METHOD"  
170 PRINT #1,"10. MODIFIED HAMMING METHOD"  
180 LOCATE 23,22  
190 INPUT "ENTER METHOD SELECTION (1-10)";MS  
200 IF MS>10 OR MS<1 THEN CLS:GOTO 20  
210 CLS  
220 INPUT" h=";H:INPUT" INITIAL VALUE OF X=";Xi:INPUT" GIVEN VALUE OF X=";Xo  
230 K=(Xo-Xi)/H:INPUT" INITIAL VALUE OF Y=";Yi:CLS  
240 ON MS GOTO 250,320,430,500,610,710,790,880,970,1120  
250 PRINT:PRINT"----- EULER METHOD -----  
260 GOSUB 1270  
270 FOR N=0 TO K-1  
280 X=Xi+N*H  
290 Yi=Yi+H*FN F(X,Yi)  
300 PRINT TAB(20),"X=";X+H,"Y(";X+H;")=";Yi  
310 NEXT:END
```

320 PRINT:PRINT "----- RUNGE-KUTTA METHOD -----"

330 GOSUB 1270

340 FOR N=0 TO K-1

350 X=Xi+N*H

360 K1=H*FN F(X,Yi)

370 K2=H*FN F(X+0.5*H,Yi+0.5*K1)

380 K3=H*FN F(X+0.5*H,Yi+0.5*K2)

390 K4=H*FN F(X+H,Yi+K3)

400 Yi=Yi+(K1+2*K2+2*K3+K4)/6

410 PRINT TAB(20),"X=";X+H,"Y(";X+H;")=";Yi

420 NEXT:END

```
430 PRINT:PRINT"----- HEUN METHOD -----"  
440 GOSUB 1270  
450 FOR N=0 TO K-1  
460 X=Xi+N*H  
470 Yi=Yi+0.5*H*(FN F(X,Yi)+FN F(X+H,Yi+H*FN F(X,Yi)))  
480 PRINT TAB(20),"X=";X+H,"Y(";X+H;" )=";Yi  
490 NEXT:END
```

500 PRINT:PRINT"----- NYSTROM METHOD -----"

510 GOSUB 1270

520 DIM Y(K+1),X(K+1):X(0)=Xi:Y(0)=Yi

530 FOR N=1 TO 2:X(N)=X(N-1)+H:NEXT:XT=X(2)

540 Y(1)=Y(0)+0.5*H*(FN F(X(0),Y(0))+FN F(X(0)+H,Y(0)+H*FN F(X(0),Y(0))))

550 PRINT TAB(20),"X=";X(1),"Y(";X(1);")=";Y(1)

560 FOR N=2 TO K

570 YT=Y(0)+2*H*FN F(X(1),Y(1))

580 PRINT TAB(20),"X=";XT,"Y(";XT;")=";YT

590 Y(0)=Y(1):Y(1)=YT:X(0)=X(1):X(1)=X(0)+H:XT=XT+H

600 NEXT:END

610 PRINT:PRINT"----- RUNGE-KUTTA-GILL METHOD -----

```
620 GOSUB 1270:A=1/SQR(2)
630 FOR N=0 TO K-1
640 X=Xi+N*H
650 K1=FN F(X,Yi)
660 K2=FN F(X+0.5*H,Yi+0.5*H*K1)
670 K3=FN F(X+0.5*H,Yi+(-0.5+A)*H*K1+(1-A)*H*K2)
680 K4=FN F(X+H,Yi-A*H*K2+(1+A)*H*K3)
690 Yi=Yi+H*(K1+2*(1-A)*K2+2*(1+A)*K3+K4)/6
700 PRINT TAB(20),"X=";X+H,"Y(";X+H;)"=";Yi:NEXT:END
710 PRINT:PRINT"----- MILNE METHOD -----
```

```
720 GOSUB 1270:GOSUB 1310
730 FOR N=4 TO K
740 P=Y(0)+4*H*(2*FN F(X(3),Y(3))-FN F(X(2),Y(2))+2*FN F(X(1),Y(1)))/3
750 YT=Y(2)+H*(FN F(X(4),P)+4*FN F(X(3),Y(3))+FN F(X(2),Y(2)))/3
760 PRINT "X=";XT-H,"PY(";XT-H;)"=";P,"Y(";XT-H;)"=";YT
770 FOR I=0 TO 2:Y(I)=Y(I+1):NEXT I:Y(3)=YT
780 FOR J=1 TO 3:X(J)=X(J+1):NEXT J:X(4)=X(3)+H:XT=XT+H:NEXT N:END
790 PRINT:PRINT"----- HAMMING METHOD -----
```

```
800 GOSUB 1270:GOSUB 1310
810 FOR N=4 TO K
820 P=Y(0)+4*H*(2*FN F(X(3),Y(3))-FN F(X(2),Y(2))+2*FN F(X(1),Y(1)))/3
830 YT=(9*Y(3)-Y(1)+3*H*(FN F(X(4),P)+2*FN F(X(3),Y(3))-FN F(X(2),Y(2))))/8
840 PRINT "X=";XT-H,"PY(";XT-H;)"=";P,"Y(";XT-H;)"=";YT
850 FOR I=0 TO 2:Y(I)=Y(I+1):NEXT I:Y(3)=YT
860 FOR J=1 TO 3:X(J)=X(J+1):NEXT J:X(4)=X(3)+H:XT=XT+H:NEXT N
870 END
```

```

880 PRINT:PRINT"----- ADAMS-MULTON METHOD -----"
890 GOSUB 1270:GOSUB 1310
900 FOR N=4 TO K
910 P=Y(3)+H*(55*FN F(X(3),Y(3))-59*FN F(X(2),Y(2))+37*FN F(X(1),Y(1))-9*FN F(X(
0),Y(0)))/24
920 YT=Y(3)+H*(9*FN F(X(4),P)+19*FN F(X(3),Y(3))-5*FN F(X(2),Y(2))+FN F(X(1),Y(1
)))/24
930 PRINT "X=";XT-H,"PY(";XT-H;")=";P,"Y(";XT-H;")=";YT
940 FOR I=0 TO 2:Y(I)=Y(I+1):NEXT I:Y(3)=YT
950 FOR J=0 TO 3:X(J)=X(J+1):NEXT J:X(4)=X(3)+H:XT=XT+H:NEXT N
960 END

```

```

970 PRINT:PRINT"----- MODIFIED MILNE METHOD -----"
980 GOSUB 1270:GOSUB 1310
990 PT=Y(0)+4*H*(2*FN F(X(3),Y(3))-FN F(X(2),Y(2))+2*FN F(X(1),Y(1)))/3
1000 Y(4)=Y(2)+H*(FN F(X(4),PT)+4*FN F(X(3),Y(3))+FN F(X(2),Y(2)))/3
1010 PRINT "X=";X(4),"PY=";PT,"MY=";" - ","Y=";Y(4)
1020 FOR N=5 TO K
1030 P=Y(1)+4*H*(2*FN F(X(4),Y(4))-FN F(X(3),Y(3))+2*FN F(X(2),Y(2)))/3
1040 M=P+28*(Y(4)-PT)/29
1050 YT=Y(3)+H*(FN F(X(5),M)+4*FN F(X(4),Y(4))+FN F(X(3),Y(3)))/3
1060 PRINT "X=";XT,"PY=";P,"MY=";M,"Y=";YT
1070 PT=P
1080 FOR I=1 TO 3:Y(I)=Y(I+1):NEXT I:Y(4)=YT
1090 FOR J=1 TO 4:X(J)=X(J+1):NEXT J:X(5)=X(4)+H:XT=XT+H
1100 NEXT
1110 END

```

```

1120 PRINT:PRINT"----- MODIFIED HAMMING METHOD -----"
1130 GOSUB 1270:GOSUB 1310
1140 PT=Y(0)+4*H*(2*FN F(X(3),Y(3))-FN F(X(2),Y(2))+2*FN F(X(1),Y(1)))/3
1150 Y(4)=Y(2)+H*(FN F(X(4),PT)+4*FN F(X(3),Y(3))+FN F(X(2),Y(2)))/3
1160 PRINT "X=";X(4),"PY=";PT,"MY=";" - ", "Y=";Y(4)
1170 FOR N=5 TO K
1180 P=Y(1)+4*H*(2*FN F(X(4),Y(4))-FN F(X(3),Y(3))+2*FN F(X(2),Y(2)))/3
1190 M=P+112*(Y(4)-PT)/121
1200 YT=(9*Y(4)-Y(2)+3*H*(FN F(X(5),M)+2*FN F(X(4),Y(4))-FN F(X(3),Y(3))))/8
1210 PRINT "X=";XT,"PY=";P,"MY=";M,"Y=";YT
1220 PT=P
1230 FOR I=1 TO 3:Y(I)=Y(I+1):NEXT I:Y(4)=YT
1240 FOR J=1 TO 4:X(J)=X(J+1):NEXT J:X(5)=X(4)+H:XT=XT+H
1250 NEXT
1260 END
1270 PRINT:PRINT TAB(10),"DIFERENTIAL EQUATION:", "y";CHR$(161);"=f(x,y)";A$
1280 PRINT:PRINT" *** BOUNDARY CONDITIONS ***"
1290 PRINT:PRINT TAB(14) "y";CHR$(161);"(";Xi;" )="Yi,"GIVEN VALUE OF X: X final
= X0, h= ;h:PRINT
1300 RETURN
1310 DIM Y(5),X(6):X(0)=Xi:Y(0)=Yi
1320 FOR N=1 TO 5:X(N)=X(N-1)+H:NEXT:XT=X(5)
1330 FOR N=1 TO 3
1340 K1=H*FN F(X(N-1),Y(N-1)):K2=H*FN F(X(N-1)+0.5*H,Y(N-1)+0.5*K1)
1350 K3=H*FN F(X(N-1)+0.5*H,Y(N-1)+0.5*K2):K4=H*FN F(X(N-1)+H,Y(N-1)+K3)
1360 Y(N)=Y(N-1)+(K1+2*K2+2*K3+K4)/6:NEXT
1370 IF MS>8 THEN 1400
1380 FOR N=1 TO 3
1390 PRINT "X=";X(N),"PY(";X(N);")=";" - " , "Y(";X(N);")=";Y(N):NEXT:RE
TURN
1400 FOR N=1 TO 3
1410 ZONE 10
1420 PRINT "X=";X(N),"PY=";" - ", "MY=";" - ", "Y=";Y(N):NEXT
1430 RETURN

```